

```

10 REM*****
20 REM*****
30 REM*****
40 REM*****          LOESUNGSVORSCHLAG ZU          *****
50 REM*****          SIGNALGENERATOR              *****
60 REM*****          UEBUNG b)                    *****
70 REM*****
80 REM*****          GST 11/88                     *****
90 REM*****
100 REM
110 REM
120 REM
130 REM
140 GOSUB Var_init: REM ***** INITIALISIERUNGSROUTINEN FUER VARIABLE, *****
150 GOSUB Ieee_init: REM ***** BUS, GENERATOR UND GRAFIK *****
160 GOSUB Dev_init
170 GOSUB Graf_init
180 REM
190 REM ***** MESSSCHLEIFE IN DER DIE FREQUENZ VON *****
200 FOR I=100 TO 1000: REM***** 100 MHZ BIS 1000 MHZ HOCHGEZAEHLT WIRD. *****
210   GOSUB Einstellung: REM ** ES WIRD DABEI JEDESMAL DER TESTPUNKT *****
220   GOSUB Messung: REM ***** ANGEWAELT UND DIE ENTSPRECHENDE TEST- *****
230   GOSUB Graf_exe: REM ***** SPANNUNG VOM GENERATOR UEBER DEN BUS AN *****
240 NEXT : REM ***** DEN CONTROLLER GESENDET. MIT GRAF-EXE *****
250 END : REM ***** WIRD DER WERT GRAFISCH DARGESTELLT. *****
260 REM
270 REM
280 REM
290 REM#####ENDE DES HAUPTPROGRAMMS#####
1000Var_init: REM=====VARIABLENDEFINITION=====
1010 REM
1020 Xl=100: Xh=1000: Yl=0: Yh=20: Lg=0: Sf=0: Co=1
1030 L$="SMH- Selbstest am "+DATUM$+" ab "+LEFT$(TIME$,5)+" Uhr"
1040 Lx$="Mhz": Ly$="Volt"
1050 REM
1060 REM ***** EINSTELLUNG DER GRAFIKPARAMETER: Xl UND Xh BZW. Yl UND *****
1070 REM ***** Yh SIND DIE ENDWERTE FUER X- UND Y- ACHSE, DIE L-STRINGS ***
1080 REM ***** SIND UEBERSCHRIFT UND ACHSENBESCHRIFTUNG.Lg, Co UND Sf *****
1090 RETURN : REM ** STELLEN SONDERFUNKTIONEN EIN. *****
1099 REM=====
1100Ieee_init: REM=====BUS- INITIALISIERUNG=====
1110 REM
1120 IEC IFC ,DCL ,TERM 13,TIME 1000
1130 REM
1140 REM
1190 RETURN
1199 REM=====
1200Dev_init: REM=====GRUNDEINSTELLUNG SMH=====
1210 REM
1220 Smh=28: REM ***** ADRESSZUWEISUNG *****
1230 Anf$="AF:OFF,AM:OFF": REM ***** ABSCHALTEN VON NF UND MODULATION *****
1240 IEC OUT Smh,Anf$
1250 REM
1260 REM
1290 RETURN
1299 REM=====

```

```

1300Einstellung: REM=====MESSEINSTELLUNG SMH=====
1310 REM *** EINSTELLUNG DER HF- FREQUENZ ***
1320 IEC OUT Smh,"RF "+STR$(I)+"M": REM *** AM GENERATOR. ***
1330 REM
1340 RETURN : REM=====
1350Messung: REM=====MESSUNG=====
1360 IEC OUT Smh,"TEST POINT 20;TEST:VOLT?;TEST OFF"
1365 IEC IN Smh,Mw$
1370 Yw=VAL(Mw$)
1380 Xw=I
1390 REM
1398 RETURN
1399 REM=====
1400 REM *****
1410 REM BEI DER MESSUNG WIRD ZUNAECHST DER TESTPUNKT 20 ANGEWAHLT, DANN
1420 REM WIRD MIT TEST:VOLT? DER WERT IM AUSGABEREGISTER BEREITGESTELLT UND
1430 REM DIE TESTFUNKTION WIEDER ABGESCHALTET.
1440 REM UEBER DEN BUS WIRD DIE GEMESSENE SPANNUNG AN DEN CONTROLLER MIT
1450 REM IEC IN UEBERSANDT. Yw UND Xw SIND DAS MESSWERTEPAAR, DAS AN DIE
1460 REM GRAFIKFUNKTION UEBERGEHEN WIRD.
1470 REM *****
1480 REM
19990 REM *****
20000 REM VARIABLENDEKLARATION (SIEHE BASIC-BESCHREIBUNG KAPITEL 4)
20005 REM *****
20010 REM XL MIN. WERT DER X-ACHSE
20020 REM XH MAX. WERT DER X-ACHSE
20030 REM YL MIN. WERT DER Y-ACHSE
20040 REM YH MAX. WERT DER Y-ACHSE
20042 REM ***** USW. *****

```